TD & TP IV
PREMIERS PAS EN SIMULATION SCILAB

1 Simulation via la méthode d'inversion.

Exercice 1. Utiliser le code Scilab suivant pour tracer l'histogramme associé à un vecteur aléatoire X dont les coordonnées sont N réalisations de variables aléatoires indépendantes et de même loi uniforme sur $[0, 1]$. On superposera la densité de la loi uniforme sur l'intervalle $[0, 1]$ et on lancera le programme pour $N = 1\ 000$, $N = 10\ 000$ jusqu'à $N = 1\ 000\ 000$. Que constatez-vous ?

```
--> clear; clf
--> N = input('Entrez la taille de l''échantillon N = ');
--> X=rand(N,1); Classe=50; histplot(Classe,X, style=2);
--> plot([0,1],[1,1],'--red');
```

Exercice 2. Utiliser le code Scilab suivant pour générer N réalisations de variables aléatoires indépendantes et de même loi exponentielle $\mathcal{E}(\lambda)$ et de Cauchy $\mathcal{C}(c)$ avec $\lambda > 0$ et $c > 0$ affectés par l'utilisateur. On tracera les moyennes empiriques successives de la loi exponentielle via la commande `cumsum` et on augmentera le nombre N de réalisations pour affiner la précision. On pourra ainsi vérifier la loi forte des grands nombres. Que se passe-t-il sur la loi de Cauchy ?

```
--> clear; clf
--> N = input('Entrez la taille de l''échantillon N = ');
--> lambda = input('Préciser la valeur du paramètre lambda = ');
--> c = input('Préciser la valeur du paramètre c = ');
--> X=rand(N,1); Y=-log(X)/lambda; Z=c*tan(%pi*(X-0.5));
--> T=[1:N]'; plot(cumsum(Y)./T,'blue');
--> plot(1/lambda*ones(N,1),'--red');
--> xgrid
```

2 Simulation via la méthode par troncature.

Exercice 3. Utiliser le code Scilab suivant pour générer N réalisations de variables aléatoires indépendantes et de même loi uniforme sur $\{1, 2, \cdots, n\}$ avec $n \geqslant 1$ et de loi géométrique $\mathcal{G}(p)$ avec $0 < p < 1$. On tracera les moyennes empiriques successives de Y afin de vérifier la loi forte des grands nombres et on effectuera le même exercice pour la loi géométrique associée à Z.

```
--> clear; clf
--> N = input('Entrez la taille de l''échantillon N = ');
--> n = input('Préciser la valeur de l''entier positif n = ');
--> p = input('Préciser la valeur du paramètre p = ');
--> X=rand(N,1);
--> Y=int(1+n*X); Z=int(1+log(X)/log(1-p));
```

3 Simulation de la loi Binomiale.

Exercice 4. Créer un code Scilab permettant de générer un vecteur aléatoire X contenant N réalisations indépendantes et de même loi Binomiale $\mathcal{B}(n,p)$ où les valeurs $N, n \geqslant 1$ et $0 < p < 1$ sont affectées par l'utilisateur. Pour N assez grand, vérifier la loi forte des grands nombres sur les moyennes empiriques successives de X.

4 Simulation via la méthode du rejet.

Exercice 5. Créer un code Scilab permettant de générer, par la méthode du rejet, N réalisations indépendantes et de même loi uniforme sur le losange de centre zéro et dont les longueurs des diagonales $a > 0$ et $b > 0$ sont affectées par l'utilisateur.

Exercice 6 Utiliser le code Scilab suivant pour simuler la loi du demi-cercle par la méthode du rejet.

```
--> clear; clf
--> N = 100000; X=[];
--> for k=1:N
-->     U=4*rand(1)-2;V=rand(1)/%pi;
-->     while (V>sqrt(4-U^2)/(2*%pi)),
-->           U=4*rand(1)-2;V=rand(1)/%pi;
-->     end;
-->     X=[X U];
--> end;
--> Classe=50; histplot(Classe,X, style=2);
--> x=[-2:0.01:2]; plot(x,sqrt(4-x.*x)/(2*%pi),'red')
--> title('Loi du demi-cercle','color','red','fontsize',4);
```

5 Simulation de la loi Normale.

Exercice 7. Utiliser l'algorithme de Box-Muller afin de générer N réalisations de variables aléatoires indépendantes et de même loi normale $\mathcal{N}(m, \sigma^2)$ où la moyenne $m \in \mathbb{R}$ et la variance $\sigma^2 > 0$ sont affectées par l'utilisateur.

```
--> clear; clf
--> N = input('Entrez la taille de l''échantillon N = ');
--> m = input('Préciser la valeur de la moyenne m = ');
--> sigma2 = input('Préciser la valeur de la variance sigma2 = ');
--> X=m*ones(N,1)+sqrt(sigma2)*sqrt(-2*log(rand(N,1))).*cos(2*%pi*rand(N,1));
--> Classe=50; histplot(Classe,X, style=2);
--> x=[m-5*sqrt(sigma2):0.01:m+5*sqrt(sigma2)];
--> plot(x,1/sqrt(2*%pi*sigma2)*exp(-(x-m).*(x-m)/(2*sigma2)),'red')
--> title('Loi Normale','color','red','fontsize',4);
```

Exercice 8. Utiliser le code Scilab suivant afin de simuler une trajectoire brownienne sur l'intervalle $[0, 1]$.

```
--> clear; clf
--> N=10000; a=0; b=1; h=(b-a)/N;
--> plot([a:h:b],cumsum(sqrt(h)*grand(N+1,1,'nor',0,1)));
--> title('Simulation d''une trajectoire Brownienne','color','red','fontsize',4);
--> xlabel('Temps'); ylabel('Valeurs du mouvement Brownien'); xgrid
```